; 这个是在于渊老师代码的基础上阉割以后可以编译成小于512字节的程序
; 旨在让程序实现对从ring0到ring3的转义，没用到的功能就删掉了。这次不得不用到栈了，2个栈我各自给了32字节，想来够用了。
; 本以为把栈缩小就够用，但是编译出来接近600字节，于是不得不对别的地方动刀了。
; 偶然发现原来的程序进入ring3以后，就进入死循环了，于是也就没有退出保护模式，回归实模式，于是code16的段和normal段也不需要了。
; 到此突发感叹，引导扇区512字节确实干不了什么事，只有操作系统通过自举，把内核载入内存，才有广阔的空间任我们驰骋。
; 很担心以后加了TSS，以及启动分页机制后，还能不能放在引导扇区，走一步看一步了。
; 调试以后，发现，stack段其实只用16字节，我们给的32字节，还给多了。

; ==========================================
; pmtest5a.asm
; 编译方法：nasm pmtest5a.asm -o pmtest5a.com
; ==========================================

%include	"pm.inc"	; 常量, 宏, 以及一些说明

%ifdef DOS_DEBUG
org 0100h
%else
org 07c00h
%endif
	jmp	LABEL_BEGIN

[SECTION .gdt]
; GDT
;                                         段基址,       段界限     , 属性
LABEL_GDT:         Descriptor       0,                 0, 0     	; 空描述符
LABEL_DESC_CODE32: Descriptor       0,  SegCode32Len - 1, DA_C + DA_32	; 非一致代码段, 32
LABEL_DESC_DATA:   Descriptor       0,       DataLen - 1, DA_DRW+DA_DPL1	; Data
LABEL_DESC_STACK:  Descriptor       0,        TopOfStack, DA_DRWA + DA_32; Stack, 32 位
LABEL_DESC_VIDEO:  Descriptor 0B8000h,            0ffffh, DA_DRW + DA_DPL3	; 显存首地址
LABEL_DESC_CODE_RING3 Descriptor 0, SegCodeRing3Len - 1, DA_C + DA_32 + DA_DPL3
LABEL_DESC_STACK3 Descriptor 0, TopOfStack3, DA_DRWA + DA_32 + DA_DPL3
; GDT 结束
GdtLen		equ	$ - LABEL_GDT	; GDT长度
GdtPtr		dw	GdtLen - 1	; GDT界限
		dd	0		; GDT基地址

; GDT 选择子
SelectorCode32		equ	LABEL_DESC_CODE32	- LABEL_GDT
SelectorData		equ	LABEL_DESC_DATA		- LABEL_GDT
SelectorStack		equ	LABEL_DESC_STACK	- LABEL_GDT
SelectorVideo		equ	LABEL_DESC_VIDEO	- LABEL_GDT
SelectorCodeRing3 equ LABEL_DESC_CODE_RING3 - LABEL_GDT + SA_RPL3
SelectorStack3 equ LABEL_DESC_STACK3 - LABEL_GDT + SA_RPL3
; END of [SECTION .gdt]

[SECTION .data1]	 ; 数据段
ALIGN	32
[BITS	32]
LABEL_DATA:
; 字符串
PMMessage:		db	"In Protect Mode now. ^-^", 0	; 进入保护模式后显示此字符串
OffsetPMMessage		equ	PMMessage - $$
DataLen			equ	$ - LABEL_DATA
; END of [SECTION .data1]


; 全局堆栈段
[SECTION .gs]
ALIGN	32
[BITS	32]
LABEL_STACK:
	times 32 db 0

TopOfStack	equ	$ - LABEL_STACK - 1
; END of [SECTION .gs]

; 堆栈段ring3
[SECTION .s3]
ALIGN 32
[BITS 32]
LABEL_STACK3:
	times 32 db 0
TopOfStack3 equ $ - LABEL_STACK3 - 1
; END of [SECTION .s3]


[SECTION .s16]
[BITS	16]
LABEL_BEGIN:
	mov	ax, cs
	mov	ds, ax
	mov	es, ax
	mov	ss, ax
	mov	sp, 0100h

	; 初始化 32 位代码段描述符
	xor	eax, eax
	mov	ax, cs
	shl	eax, 4
	add	eax, LABEL_SEG_CODE32
	mov	word [LABEL_DESC_CODE32 + 2], ax
	shr	eax, 16
	mov	byte [LABEL_DESC_CODE32 + 4], al
	mov	byte [LABEL_DESC_CODE32 + 7], ah

	; 初始化数据段描述符
	xor	eax, eax
	mov	ax, ds
	shl	eax, 4
	add	eax, LABEL_DATA
	mov	word [LABEL_DESC_DATA + 2], ax
	shr	eax, 16
	mov	byte [LABEL_DESC_DATA + 4], al
	mov	byte [LABEL_DESC_DATA + 7], ah

	; 初始化堆栈段描述符
	xor	eax, eax
	mov	ax, ds
	shl	eax, 4
	add	eax, LABEL_STACK
	mov	word [LABEL_DESC_STACK + 2], ax
	shr	eax, 16
	mov	byte [LABEL_DESC_STACK + 4], al
	mov	byte [LABEL_DESC_STACK + 7], ah
	
	xor eax, eax
	mov ax, ds
	shl eax, 4
	add eax, LABEL_CODE_RING3
	mov word [LABEL_DESC_CODE_RING3 + 2], ax
	shr eax, 16
	mov byte [LABEL_DESC_CODE_RING3 + 4], al
	mov byte [LABEL_DESC_CODE_RING3 + 7], ah
	
	; 初始化堆栈段描述符(Ring3)
	xor	eax, eax
	mov	ax, ds
	shl	eax, 4
	add	eax, LABEL_STACK3
	mov	word [LABEL_DESC_STACK3 + 2], ax
	shr	eax, 16
	mov	byte [LABEL_DESC_STACK3 + 4], al
	mov	byte [LABEL_DESC_STACK3 + 7], ah

	; 为加载 GDTR 作准备
	xor	eax, eax
	mov	ax, ds
	shl	eax, 4
	add	eax, LABEL_GDT		; eax <- gdt 基地址
	mov	dword [GdtPtr + 2], eax	; [GdtPtr + 2] <- gdt 基地址

	; 加载 GDTR
	lgdt	[GdtPtr]

	; 关中断
	cli

	; 打开地址线A20
	in	al, 92h
	or	al, 00000010b
	out	92h, al

	; 准备切换到保护模式
	mov	eax, cr0
	or	eax, 1
	mov	cr0, eax

	; 真正进入保护模式
	jmp	dword SelectorCode32:0	; 执行这一句会把 SelectorCode32 装入 cs, 并跳转到 Code32Selector:0  处

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

; END of [SECTION .s16]


[SECTION .s32]; 32 位代码段. 由实模式跳入.
[BITS	32]

LABEL_SEG_CODE32:
	mov	ax, SelectorData
	mov	ds, ax			; 数据段选择子
	mov	ax, SelectorVideo
	mov	gs, ax			; 视频段选择子

	mov	ax, SelectorStack
	mov	ss, ax			; 堆栈段选择子

	mov	esp, TopOfStack


	; 下面显示一个字符串
	mov	ah, 0Ch			; 0000: 黑底    1100: 红字
	xor	esi, esi
	xor	edi, edi
	mov	esi, OffsetPMMessage	; 源数据偏移
	mov	edi, (80 * 10 + 0) * 2	; 目的数据偏移。屏幕第 10 行, 第 0 列。
	cld
.1:
	lodsb
	test	al, al
	jz	.2
	mov	[gs:edi], ax
	add	edi, 2
	jmp	.1
.2:	; 显示完毕
	
	push SelectorStack3
	push TopOfStack3
	push SelectorCodeRing3
	push 0
	retf
	
	jmp $

SegCode32Len	equ	$ - LABEL_SEG_CODE32
; END of [SECTION .s32]


; END of [SECTION .s16code]

[SECTION .ring3]
ALIGN 32
[BITS 32]
LABEL_CODE_RING3:
	mov ax, SelectorVideo
	mov gs, ax
	
	mov edi, (80 * 14 + 0) * 2
	mov ah, 0Ch
	mov al, '3'
	mov [gs:edi], ax
	
	jmp $
SegCodeRing3Len equ $ - LABEL_CODE_RING3

